



Agriculture and  
Agri-Food Canada

Agriculture et  
Agroalimentaire Canada




# Using PROC GMAP to Display Data from a Population Genetics Simulation Model

Marjorie Smith and Sheila Woods

Canada

# Outline

- The model
- The data
- Graphics devices and output destinations
- Creating the map file
- Setting up the graphics environment
- Mapping the response data



## The Problem:



Wheat midge, a tiny insect that destroys wheat kernels



## The Solution:

- Resistant wheat that contains a gene that kills the larvae

But:

- A few insects can survive, so the population eventually adapts to the resistant wheat

## Spatial Simulation Model:

- Developed to explore management strategies so the resistant wheat remains effective against the insect population

# Output Data from the Model

- A wheat growing area is represented by a grid of 25 x 25 farms
- Each observation represents data for one farm
- There are six variables, written to a file as tab-delimited text:
  - Insect generation
  - Row and column (location of a farm)
  - Type of wheat grown (resistant or susceptible)
  - Frequency of adaptive trait in the insect population
  - Insect population density

# An Example of an Output Data File

Input filename: in\_snr005\_1\_06.txt

1 dispersals per generation

proportion of females dispersing is 0.7

Control threshold: 7500 Efficacy: 0.8

0 0 0 RR 0.005 1300

0 0 1 RR 0.005 1300

0 0 2 RR 0.005 1300

0 0 3 RR 0.005 1300

. . . *more data lines* . . .

100 24 20 RR 0.996 9342.4551

100 24 21 SS 0.8958 11936.896

100 24 22 SS 0.8715 11180.634

100 24 23 SS 0.9175 10018.127

100 24 24 RR 0.997 5483.3527

## Specifying the device (graphics format) and output destination

- gif files viewed in SAS Viewer or a web browser:  

```
goptions device=gif rotate=landscape;  
ods html body="index.html" path="subdirectory";
```
- cgm files imported into a MS Office application (one graph per file):  

```
goptions device=cgmof971 gsfname=cgm1;  
filename cgm1=".";
```
- pdf file viewed in Adobe Acrobat (all graphs in one file):  

```
goptions device=pdf;  
ods pdf file="filename.pdf";
```
- Graphs can also be sent directly to a printer


# Using CGM Format for Flexibility

cgm (computer graphics metafile) is very useful for displaying maps that can be edited

- Must be imported (to MS Word, Excel or PowerPoint) to be viewed
- High resolution
- Individual parts of the map can be edited in PowerPoint; for example:
  - The map, legend and text can be rearranged
  - The legend can be edited separately from the map
  - Text can be edited, moved, added or deleted

# Using a Map to Display the Data


- Example: insect population density in wheat of each of the 625 farms
- This is a choropleth map, a 2D map that uses colour and fill pattern combinations to represent different categories or levels of magnitude



- This is a gif file
- Output destination is html – can be viewed in a web browser

# Using a Map to Display the Data

- In this example, the map was saved as a cgm file
- Imported and edited in PowerPoint (can also import to MS Word or Excel)




# Creating the Map File

- creates the map as a grid of square cells

```
data a;  
    /* get data values to make the map */  
    infile 'out_snr_08_60_1.txt' dlm='09'x dsd  
    firstobs=5;  
    input gen row column crop $ vfreq npop;  
    label gen='generation';  
  
data perm.smap_v2b;  
    set a;  
    x=row; y=column; output;  
    x+1; output;  
    y+1; output;  
    x+(-1); output;  
    keep gen row column x y;
```

# Example Showing Part of a Map File

- gives the coordinates for the corners of each square



The screenshot shows the SAS System Viewer interface for a file named 'smap\_v2b.sas7bdat'. The window title is 'SAS System Viewer - smap\_v2b.sas7bdat'. The menu bar includes 'File', 'Edit', 'View', 'Window', and 'Help'. The toolbar contains various icons for file operations and data manipulation. The main data window, titled 'smap\_v2b.sas7bdat', displays a table with the following data:

	gen	row	column	x	y
1	0	0	0	0	0
2	0	0	0	1	0
3	0	0	0	1	1
4	0	0	0	0	1
5	0	0	1	0	1
6	0	0	1	1	1
7	0	0	1	1	2
8	0	0	1	0	2
9	0	0	2	0	2
10	0	0	2	1	2
11	0	0	2	1	3
12	0	0	2	0	3
13	0	0	3	0	3
14	0	0	3	1	3

# Setting up the Graphics Environment

```
ods listing close;  
x "md s60_graphs";  
ods html body="index.html" path="s60 graphs"  
  (url=none);  
goptions cback=grayc1 device=gif rotate=landscape;
```

```
legend1 across=1  
  position=(middle right)  
  offset=(-2 cm)  
  label=(position=top 'q midpoint')  
  shape=bar(6,1);
```

```
legend2 across=1  
  position=(middle right)  
  offset=(-2 cm)  
  label=(position=top 'females / 10 m2 midpoint')  
  shape=bar(6,1);
```

## Setting up the Graphics Environment (cont.)

```
pattern1 v=msolid c=grayff;  
pattern2 v=msolid c=graye8;  
pattern3 v=msolid c=grayd8;  
pattern4 v=msolid c=grayc8;  
pattern5 v=msolid c=grayb8;  
pattern6 v=msolid c=graya8;  
pattern7 v=msolid c=gray94;  
pattern8 v=msolid c=gray7d;  
pattern9 v=msolid c=gray68;  
pattern10 v=msolid c=gray48;  
pattern11 v=msolid c=gray2f;
```

Each pattern is used for one midpoint value, so the number of patterns must match the number of midpoints for the data

The patterns are used sequentially and in rotation

Grayscale values can go from grayff (white) to gray00 (black)

I used “trial and error” to get the range of grayscale values for the map

# Mapping the Response Data

```
data a;
  /* data values to be plotted on the map */
  infile 'out_snr60_02_04_1.txt' dlm='09'x dsd
  firstobs=4;
  input gen row column crop $ vfreq npop;
  if gen in (1, 19, 29);
  label gen='generation';

proc sort;
  by gen;
```

In this example, data for three selected generations will be plotted

Select the same generations from the map file when plotting in  
PROC GMAP


## Mapping the Response Data (cont.)

```
proc gmap data=a map=perm.smap_v2b (where= (gen in
  (1, 19, 29)));
  id row column;
  by gen;
  choro vfreq /midpoints=0 to 1 by .1
  legend=legend1;
  choro npop /midpoints=0 to 10000 by 1000
  legend=legend2;
  title 'WM v2b - data from
  out_snr60_02_04_1.txt';
```

id statement – list variables common to the map and response data sets, to match response values with map areas

Check that the number of midpoints matches the number of patterns

# Create separate directories for each set of maps



- Maps created in a PROC GMAP step are numbered sequentially
- The body file (index.html) is used to view the maps
- If you run another program to create different maps in the same subdirectory, SAS will overwrite the maps using the same numbering system, so create a new sub-directory for each set of maps



# Canada